
第一部分 上机实训 

实训一 熟悉 Visual Basic 6.0 编程环境 

实训目的 

（1）掌握 Visual Basic的启动与退出方法。 

（2）掌握建立、编辑和运行一个简单的 Visual Basic 应用程序的步骤和过程。 

实训内容 

【实训 1.1】 

创建一个新工程。建立的用户界面如图 111 所示。程序运行时，单击窗体可以改变标签 

中的文字颜色。然后以 mypro_1.frm和 mypro_1.vbp 文件名将编写的程序保存在磁盘上。 

图 111  实训 1.1的应用程序界面 

实训步骤： 

（1）启动 Visual Basic 6.0。启动 Visual Basic 6.0 的方法有多种，通常可以双击桌面上的 

快捷图标进行启动。 

（2）启动 Visual Basic 6.0 后，在系统弹出的“新建工程”对话框中，询问要创建的工程 

类别，系统默认的类别是创建一个“标准  EXE”应用程序。在此，直接单击“打开”按钮， 

进入编写程序的状态。 

（3）现在就可以改变系统提供的窗体 Form1 的属性。注意，此时窗体 Form1 的四周有 
8 个小方框，表示窗体对象处于被选中状态。8 个小方框中有 3 个是实心蓝色的，将光标移 

动到这些实心小方框上，按住鼠标移动，就可以改变窗体的大小。改变窗体的大小其实就是 

改变窗体的 Height 和 Width 属性。因此，也可以直接在属性窗口中对这两个属性进行修改， 

例如可以把窗体的 Height 值设为 3000，Width 值设为 5000，还可以把窗体的 Caption  属性 

设为“练习 1” 。 

（4）向窗体 Form1 中增加一个标签控件（Label1），并用鼠标将其选中（选中时，周边有 
8 个蓝色小方块）。在属性窗口中将 Label1 的 BorderStyle 属性值设为 1，BackColor 属性设为 

白色，Caption属性设为“我的 Visual Basic程序” 。在属性窗口中找到 Font 属性，单击其右侧

的 按钮，在弹出的“字体”对话框中可以对标签的字体、字号进行设置。在此，可以选择 

“宋体” 、三号字，再单击“确定”按钮，关闭“字体”对话框。 

（5）为了在程序运行过程中让文字改变颜色，可以对窗体编写一段单击事件过程，具体 

方法是：双击窗体，打开代码编辑窗口（见图 112），在过程框中选择单击事件 Click，代码 

窗口中出现相应的 Form_Click( )过程框架： 
Private Sub Form_Click() 

End Sub 
在上述两行代码之间添加如下代码： 

Label1.ForeColor = QBColor(Int(Rnd * 16)) 
这条语句的作用是，标签上的文本颜色属性由一个随机数来确定，这样每次单击窗体时， 

文字的颜色就会不同。语句中使用了几个内部函数，有关的介绍可参考教材。 

图 112  代码编辑窗口 

（6）单击工具栏中的启动按钮 或按 F5 键运行程序。如果程序编写正确，屏幕上就会 

出现相对于图 111 所示的窗体运行界面，单击窗体，标签上的文字就会改变颜色。如果程序 

编写有错误，系统会给出错误信息，可以按照提示修改程序，然后再运行。若要结束程序的运 

行，可以单击工具栏上的终止运行按钮 。 

（7）程序能正确运行后，应及时保存。注意，一个 Visual  Basic 工程至少包含两个磁盘 

文件，应分别进行保存。选择“文件”菜单中的“保存工程”命令，系统弹出如图 113 所示 

的“文件另存为”对话框，用于保存窗体文件。此时，可以选择文件的存放位置和文件名。建 

议不使用系统提供的默认文件名，而是输入自己指定的文件名。这里，在“文件名”文本框中 

输入 mypro_1，单击“保存”按钮。保存窗体文件后，系统又弹出“工程另存为”对话框，用 

于保存工程文件。同样，在“文件名”文本框中输入mypro_1，然后单击“保存”按钮。这样 

就完成了保存该工程文件的任务。以后要在 Visual Basic环境中打开该工程时，直接打开工程 

文件 mypro_1.vbp即可。 

（8）前面保存的工程文件只能在 Visual  Basic 编程环境中运行。如果想脱离编程环境直 

接运行该程序，就需要将其生成可执行文件。选择“文件”菜单中的“生成 mypro_1.exe”命 

令，系统弹出“生成工程”对话框，确认要生成的应用程序文件名为 mypro_1.exe以及该文件 

的存放位置后，单击“确定“按钮，系统就会按照指定的文件名和路径生成可独立运行的 EXE 
文件。

（9）退出 Visual Basic 系统。退出的方法有多种，可以在“文件”菜单中选择“退出” 

命令退出。一般情况下，系统在退出前会提示是否保存工程文件，此时根据需要进行选择 

即可。



3 

第一部分 上机实训 

图 113  “文件另存为”对话框 

注意事项： 

（1）L 的小写字母“l”和数字“1”的形状几乎相同，在输入程序代码时要注意，千万 

不能混淆。 实训步骤 （5） 中输入的代码 “Label1.ForeColor = QBColor(Int(Rnd * 16))” 中的 Label1， 

第 5 个字符是英文字母“L”的小写，第 6 个字符是数字“1” 。 

（2）双击窗体打开代码编辑窗口时，系统提供的是窗体装载过程框架。 
Private Sub Form_Load() 

End Sub 
注意：不要把代码直接加入该框架，而是选择单击事件过程框架。 

【实训 1.2】 

编程实现：程序开始运行时，窗体上的文本框显示： “欢迎使用我的 VB 程序” ；当用户单 

击（Click）窗体 Form1 时，文本框显示“你单击了窗体” ；当用户双击（DblClick）窗体 Form1 
时，文本框显示“你双击了窗体” ，如图 114 所示。 

图 114  实训 1.2的程序运行界面 

提示：本实训的目的是理解窗体的装载（Form_Load）、单击（Click）和双击（DblClick） 

事件。

实训步骤： 

（1） 建立一个新工程， 在窗体上添加一个文本框控件 Text1， 并将其 Height 属性设为 700， 
Width属性设为 3300，然后通过 Font 属性把字体设为宋体、三号字。

（2）编写窗体的 Form_Load事件过程代码。 
Private Sub Form_Load() 

Text1.Text = "欢迎使用我的 VB程序" 
End Sub 

（3）编写窗体的 Click事件过程代码。 
Private Sub Form_Click() 

Text1.Text = "你单击了窗体" 
End Sub 

（4）编写窗体的 DblClick事件过程代码。 
Private Sub Form_DblClick() 

Text1.Text = "你双击了窗体" 
End Sub 

（5）调试、运行程序。 

【实训 1.3】 

建立一个新工程，练习标签、文本框和命令按钮的使用。 

实训步骤： 

（1） 在窗体上添加一个标签 （Label1）、 一个文本框 （Text1） 和两个命令按钮 （Command1、 
Command2）。将文本框的 Font 属性设置为“隶书、四号” ，Text 属性设置为“欢迎您学习 VB” ； 

将两个命令按钮的 Caption属性分别设为“前景色”和“背景色” ，如图 115 所示。 

（2）双击命令按钮 Command1，进入代码编辑窗口。编写命令按钮 Command1 的 Click 
事件（前景色）过程代码。 

Private Sub Command1_Click() 
Label1.Caption = "文本框的前景色为红色" 
Text1.ForeColor = vbRed 

End Sub 
（3）编写命令按钮 Command2 的 Click事件（背景色）过程代码。 

Private Sub Command2_Click() 
Label1.Caption = "文本框的背景色为蓝色" 
Text1.BackColor = vbBlue 

End Sub 
（4）调试、运行程序，其运行结果如图 116 所示。 

图 115  实训 1.3的窗体设计界面 图 116  实训 1.3的窗体运行界面

实训二 程序设计的基本概念 

实训目的 

（1）掌握数据类型、表达式以及赋值语句的书写规则。 

（2）掌握常用内部函数的使用方法。 

（3）掌握 InputBox 和 MsgBox 的使用方法。 

（4）掌握 Print 方法以及与之相关的函数的使用方法。 

实训内容 

【实训 2.1】 

编写一个应用程序，初始界面如图 121 所示。程序运行时，单击“开始”按钮，弹出如 

图 122 所示的对话框，要求用户输入一个任意的角度值，单击“确定”按钮后程序根据输入 

的数据把相关的三角函数值按一定的格式输出到窗体上。程序的运行结果如图 123 所示。 

图 121  实训 2.1的程序初始界面 图 122 “角度输入”对话框 

图 123  实训 2.1的运行结果 

实训步骤： 

（1）启动 Visual Basic 6.0 后，创建一个“标准 EXE”应用程序，进入程序编辑状态。 

（2）修改窗体的尺寸，并把窗体的 Caption 属性设为“三角函数计算” ，再向窗体 Form1 
中增加一个命令按钮控件（Command1），并将其 Caption属性设为“开始” 。 

（3）双击命令按钮 Command1，进入代码编辑窗口。编写 Command1 的单击事件过程 

如下： 
Private Sub Command1_Click()

Dim x As Single, a As Single 
x = Val(InputBox("请输入一角度的度数值", "角度输入")) 
a = 3.1415926 / 180 * x  '将度数转换为弧度 
Print 
Print Tab(4); "角度"; Tab(14); "弧度"; Tab(28); "Sin(x)"  '输出表头 
Print Tab(4); Format(x, "###.0"); Tab(14);  _ 
Format(a, "0.0000"); Tab(28); Format(Sin(a), "0.0000")  '续行，输出计算结果 

End Sub 
（4）输入以上程序后，单击工具栏中的启动按钮 或按 F5 键运行程序。如果程序编写 

正确，单击“开始”按钮后，就会出现如图 122 所示的输入数据提示框。输入数据后，单击 

“确定”按钮，得到如图 123 所示的结果。若要结束程序的运行，可以单击工具栏上的结束 

运行按钮 。 

注意事项： 

（1）在输入程序时，要特别注意程序中起分界符作用的括号、双引号、分号、逗号等均必 

须使用西文符号。一般情况下，中文字符（包括中文标点符号）只能用于给字符串或对象名赋值。 

（2）InputBox 函数的返回值类型是字符型，所以在程序中用 Val(  )函数将其转换为数值 

型数据后，再赋值给变量 a。 

（3）Visual Basic中所有的三角函数的参数（自变量）均要求使用弧度值，不能直接用度 

数代入计算。 

（4）如果某条语句太长，可以使用续行符“  _” （一个空格紧跟一下划线）将其分为多 

行书写。 

（5）Print 语句与 Tab(  )、Format(  )函数联合使用可以使得输出的数据整齐美观，在调试 

程序时，可以比较一下不使用这两个函数的输出效果。 

【实训 2.2】 

假设某电话系统按秒计费（每秒 0.02 元），编程实现其通话计费程序。要求能够显示通话 

开始时间、结束时间、通话时间以及通话费用。 

实训步骤： 

（1）建立一个新工程，在窗体上添加四个标签、四个文本框和两个命令按钮，并参考图 
124设置有关对象的属性。 

图 124  实训 2.2的窗体设计图 

（2）双击窗体进入代码编辑窗口，在代码编辑窗口右上角的下拉列表框中选择“通用”

（见图  125 所示），目的是说明几个模块级变量供几个事件过程共同使用。在代码编辑窗口 

的第一行输入变量声明语句： 
Dim t1 As Date, t2 As Date, t3 As Date 

该语句定义了三个日期型变量 t1、t2、t3。有关变量作用域的内容，在主教材第 6 章进行 

详细说明。 

图 125  选择“通用”声明区声明变量 

（3）编写命令按钮 Command1 的 Click事件（通话开始事件）过程代码。 
Private Sub Command1_Click() 

t1 = Time                        '读取通话开始时间，并存入变量 t1中 
Text1.Text = t1                '显示通话开始时间 
Text2.Text = ""                '使其余文本框不显示内容 
Text3.Text = "" 
Text4.Text = "" 

End Sub 
（4）编写命令按钮 Command2 的 Click事件（通话结束事件）过程代码。 

Private Sub Command2_Click() 
t2 = Time                                      '读取通话结束时间，并存入变量 t2中 
Text2.Text = t2                              '显示通话结束时间 
t3 = t2  t1                                      '计算实际通话时间 
Text3.Text = t3                              '显示通话时间 
Text4.Text = Second(t3) * 0.02    '计算通话费用，并显示在文本框 4中 

End Sub 
（5）调试、运行程序，其运行效果如图 126 所示。 

图 126  运行结果 

注意事项： 

（1） 文本框控件的主要功能是输入或输出显示一个文本数据， 它与其他大部分控件不同，

没有 Caption属性。它与标签控件的主要区别是，后者一般仅用于输出显示文本。 

（2）Time 函数的功能是读取计算机系统的当时时间，它的返回值是  Date 类型，即： 
#hh:mm:ss#。程序在计算通话费用时，用 Second(  )函数把 Date类型的数据转化成秒数后，再 

计算通话费用，并将其显示在一个文本框中。 

【实训 2.3】 

创建一个应用程序，其用户登录界面如图 127 所示，要求用户输入姓名。当用户输入姓 

名，并单击“确定”按钮后，系统会弹出如图 128所示的欢迎界面。 

图 127  应用程序界面 

图 128  实训 2.3的程序运行结果 

实训步骤： 

（1）建立一个新工程，在窗体上添加两个标签、一个文本框和一个命令按钮，并参考图 
127设置有关对象的属性。 

（2）双击窗体进入代码编辑窗口，编写窗体装载事件过程。 
Private Sub Form_Load() 

Text1.Text = "" 
End Sub 

该过程的作用是程序开始运行时，首先清空文本框中的文字，便于用户输入。 

（3）编写命令按钮 Command1 的单击事件过程。 
Private Sub Command1_Click() 

MsgBox "欢迎你：" & Text1.Text & "同学！", vbOKOnly, "实训 2.3" 
End Sub 

（4）调试、运行程序，其运行结果如图 128 所示。 

注意事项： 

（1）消息框 MsgBox 常用于在程序运行过程中显示一些提示性的消息。它有两种用法： 

语句方式和函数方式，上述程序中使用的是语句方式，如果使用函数方式，则应写成： 
Dim a As Integer 
a = MsgBox("欢迎你：" & Text1.Text & "同学！", vbOKOnly, "实训 2.3") 

（2）程序运行时，当用户单击“确定”按钮后，变量 a 得到整型值 1。使用语句方式时，

没有返回值。 

【实训 2.4】 

利用 InputBox 输入三角形三条边的长度 a、b、c，计算并显示三角形的面积。计算三角形 

面积的公式为： 
area s(s a)(s b)(s c) = − − − ，其中s (a b c) / 2 = + + 

实训步骤： 

（1）建立一个新工程，可适当调整窗体的大小，但不添加任何控件，将窗体的  Caption 
属性修改为“三角形面积计算” 。 

（2）双击窗体进入程序代码编辑窗口，编写窗体的单击事件过程如下： 
Private Sub Form_Click() 

Dim a As Single, b As Single, c As Single 
Dim s As Single, area As Single 
a = Val(InputBox("请输入边长 a：")) 
b = Val(InputBox("请输入边长 b：")) 
c = Val(InputBox("请输入边长 c：")) 
s = (a + b + c) / 2 
area = Sqr(s * (s  a) * (s  b) * (s  c)) 
Print Tab(5); "a"; Tab(13); "b"; Tab(21); "c"; Tab(29); "area" 
Print Tab(4); a; Tab(12); b; Tab(20); c; Tab(28); area 

End Sub 
注意事项： 

（1）在书写数学表达式时，乘号“*”一定不能省略。 

（2）注意程序中两条  Print 语句的区别，前一句是输出字符串本身，后一句是输出变 

量的值。

实训三 选择结构程序设计 

实训目的 

（1）熟练掌握 If语句的单分支、双分支和多分支形式的使用方法。 

（2）熟练掌握多分支 Select Case语句的使用方法。 

（3）掌握嵌套的 If语句的使用方法。 

实训内容 

【实训 3.1】 

从键盘输入一个正整数作为年份，编程判断该年是否为闰年。程序的初始设计界面如图 
131 所示；程序运行时，单击“判断”按钮，弹出如图  132 所示的对话框，程序的运行结 

果如图 133 所示。 

图 131  实训 3.1的程序初始界面 

图 132  “数据输入”对话框 图 133  实训 3.1的程序运行结果 

实训步骤： 

（1）启动 Visual Basic 6.0 后，创建一个“标准 EXE”应用程序，进入窗体设计界面。 

（2）修改窗体的尺寸，并把窗体的 Caption 属性值设为“判断闰年” ，然后向窗体 Form1 
中增加一个命令按钮控件（Command1），并将其 Caption属性设为“判断” 。 

（3）双击命令按钮 Command1，进入代码编辑窗口，编写 Command1 的单击事件过程， 

代码如下： 
Private Sub Command1_Click() 

Dim year As Integer 
year = Val(InputBox("请输入年份")) 
If year Mod 400 = 0 Or (year Mod 4 = 0 And year Mod 100 <> 0) Then


